Dataplot Vol 1 Vol 2

# PAGE TEST

Name:
PAGE TEST
Type:
Analysis Command
Purpose:
Perform a Page test that k treatments have identical effects against the alternative of ordered treatment effects.
Description:
The Page test is a non-parametric test for analyzing randomized complete block designs. It is derived from a Spearman's rho rank correlation of the Friedman within block rankings and the specified ordering of the treatments. The Page test statistic is actually a monotonic function of Spearman's rho when there is no ties in the data.

The distinction between the Friedman test and the Page test is that the Page test is used when you want to test for ordered effects. By ordered effects, we mean that treatment 1 is less than or equal to treatment 2 which is less than or equal to treatment 3 and so on (and at least one of these should be strictly less than).

As with the Friedman test, assumes that there are k experimental treatments (k >= 2). The observations are arranged in p blocks, that is

Treatment
Block 1 2 ... k
1 X11 X12 ... X1k
2 X21 X22 ... X2k
3 X31 X32 ... X3k
... ... ... ... ...
b Xb1 Xb2 ... Xbk

Let R(Xij) be the rank assigned to Xij within block i (i.e., ranks within a given row). Average ranks are used in the case of ties. The ranks are summed to obtain

$$R_{j} = \sum_{i=1}^{b}{R(X_{ij})}$$

Then the Page test is

 H0: The treatment effects have identical effects Ha: Ha: $$\mu_1 \le \mu_2 \le \ldots \le \mu_k$$ Test Statistic: The Page test statistic is $$\mbox{T4} = \sum_{j=1}^{k}{j R_{j}}$$ In order to compare this to a standard normal distribution, this statistic is transformed to $$\mbox{T5} = \frac{T4 - bk(k+1)^2/4} {\sqrt{b(k^3 - k)^2/(144(k-1))}}$$ Significance Level: $$\alpha$$ Critical Region: $$\mbox{T5} > \Phi^{-1}(\alpha)$$ where $$\Phi^{-1}$$ is the normal percent point function. Conclusion: Reject the null hypothesis if the test statistic is in the critical region.

Syntax:
PAGE TEST <y> <block> <treat>             <SUBSET/EXCEPT/FOR qualification>
where <y> is the response variable;
<block> is a variable that identifies the block;
<treat> is a variable that identifies the treatment;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.
Examples:
PAGE TEST Y BLOCK TREATMENT
PAGE TEST Y X1 X2
PAGE TEST Y BLOCK TREATMENT SUBSET BLOCK > 2
Note:
Dataplot formulates the test as a "less than or equal to" test. That is, the treatment that is hypothesized to be the smallest should be coded as 1, the treatment that is hypothesized to be the second smallest should be coded as 2, and so on. If your treatments are coded in the opposite order, you can reverse them with the command

LET TREAT2 = FLIP TREAT
Note:
In Dataplot, the variables should be given as:

Y BLOCK TREAT

X11 1 1
X12 1 2
... 1 ...
X1k 1 k
X21 2 1
X22 2 2
... 2 ...
X2k 2 k
... ... ...
Xb1 b 1
Xb2 b 2
... b ...
Xbk b k

If your data are in a format similar to that given in the DESCRIPTION section (i.e., you have colums Y1 to Yk, each with b rows), you can convert it to the format required by Dataplot with the commands:

LET NBLOCK = SIZE Y1
LET BLOCKID = SEQUENCE 1 1 NBLOCK
LET Y BLOCK TREAT = REPLICATED STACK Y1 Y2 Y3 Y4 Y5 BLOCKID
PAGE TEST Y BLOCK TREAT
Note:
The response, ranked response, block, and treatment are written to the file dpst1f.dat in the current directory.
Note:
The Page test is based on the following assumptions:

1. The b rows are mutually independent. That means that the results within one block (row) do not affect the results within other blocks.

2. The data can be meaningfully ranked.
Note:
The following statistics are also supported:

LET A = PAGE TEST Y X1 X2
LET A = PAGE TEST NORMALIZED Y X1 X2
LET A = PAGE TEST CDF Y X1 X2
LET A = PAGE TEST PVALUE Y X1 X2

The PAGE TEST returns the T4 value described above and the PAGE TEST NORMALIZED returns the T5 value described above.

Enter HELP STATISTICS to see what commands can use these statistics.

Default:
None
Synonyms:
None
Related Commands:
 FRIEDMAN TEST = Perform a Friedman test. QUADE TEST = Perform a Quade test. ANOVA = Perform an analysis of variance. DURBIN TEST = Perform a Durbin test for two-way incomplete balanced block designs. COCHRAN TEST = Perform a Cochran test for two-way complete block designs (binary data). KRUSKAL WALLIS = Perform a Kruskall Wallis test. BLOCK PLOT = Generate a block plot. DEX SCATTER PLOT = Generates a dex scatter plot. DEX ... PLOT = Generates a dex plot for a statistic.
Reference:
Conover, "Practical Nonparametric Statistics," Third Edition, Wiley, 1999, pp. 380-383.
Applications:
Analysis of Variance
Implementation Date:
2013/2
Program:

.  Test Page test as given in
.  Conover (1999), "Practical Non-Parametric Statistics"
.  Third Edition, pp. 380-381.
.
79   1    1
78   1    2
72   1    3
66   1    4
75   1    5
64   1    6
68   1    7
70   1    8
76   2    1
76   2    2
74   2    3
65   2    4
75   2    5
66   2    6
70   2    7
72   2    8
77   3    1
76   3    2
77   3    3
68   3    4
72   3    5
65   3    6
68   3    7
78   3    8
84   4    1
80   4    2
78   4    3
72   4    4
74   4    5
69   4    6
74   4    7
76   4    8
82   5    1
80   5    2
75   5    3
65   5    4
77   5    5
68   5    6
70   5    7
77   5    8
end of data
.
let statval = page test             y block treat
let statva2 = page test normalized  y block treat
let cdf     = page test cdf         y block treat
let pval    = page test pvalue      y block treat
.
set write decimals 4
page test y block treat

The following output is generated.

PARAMETERS AND CONSTANTS--

STATVAL --       401.5000
STATVA2 --         2.9345
CDF     --         0.9983
PVAL    --         0.0017

Page Two Factor Test

Response Variable: Y
Group-ID Variable (Block): BLOCK
Group-ID Variable (Treatment): TREAT

H0: u1 = u2 = ... = uk
Ha: u1 <= u2 <= ... <= uk

Summary Statistics:
Total Number of Observations:                        40
Number of Blocks:                                     8
Number of Treatments:                                 5

Test:
Page Test Statistic:                           401.5000
Page Normalized Test Statistic:                  2.9344
CDF of Test Statistic:                           0.9983
P-Value:                                         0.0016

Percent Points of the Normal Reference Distribution
-----------------------------------
Percent Point               Value
-----------------------------------
50.0    =          0.000
75.0    =          0.674
90.0    =          1.282
95.0    =          1.645
97.5    =          1.960
99.0    =          2.326
99.9    =          3.089

Conclusions (Upper 1-Tailed Test)
----------------------------------------------
Alpha    CDF   Critical Value     Conclusion
----------------------------------------------
10%    90%            1.282      Reject H0
5%    95%            1.645      Reject H0
2.5%  97.5%            1.960      Reject H0
1%    99%            2.326      Reject H0



NIST is an agency of the U.S. Commerce Department.

Date created: 03/08/2013
Last updated: 11/04/2015